
G08 – Nonparametric Statistics

G08RAF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

G08RAF calculates the parameter estimates, score statistics and their variance-covariance matrices for
the linear model using a likelihood based on the ranks of the observations.

2 Specification

SUBROUTINE G08RAF(NS, NV, NSUM, Y, IP, X, NX, IDIST, NMAX, TOL,
1 PARVAR, NPVAR, IRANK, ZIN, ETA, VAPVEC, PAREST,
2 WORK, LWORK, IWA, IFAIL)
INTEGER NS, NV(NS), NSUM, IP, NX, IDIST, NMAX, NPVAR,
1 IRANK(NMAX), LWORK, IWA(NMAX), IFAIL
real Y(NSUM), X(NX,IP), TOL, PARVAR(NPVAR,IP),
1 ZIN(NMAX), ETA(NMAX), VAPVEC(NMAX∗(NMAX+1)/2),
2 PAREST(4∗IP+1), WORK(LWORK)

3 Description

Analysis of data can be made by replacing observations by their ranks. The analysis produces inference
for regression parameters arising from the following model.

For random variables Y1, Y2, . . . , Yn we assume that, after an arbitrary monotone increasing differentiable
transformation, h(.), the model

h(Yi) = x
T
i β + εi (1)

holds, where xi is a known vector of explanatory variables and β is a vector of p unknown regression
coefficients. The εi are random variables assumed to be independent and identically distributed with
a completely known distribution which can be one of the following: Normal, logistic, extreme value or
double-exponential. In Pettitt [1] an estimate for β is proposed as β̂ =MXTa with estimated variance-
covariance matrix M . The statistics a and M depend on the ranks ri of the observations Yi and the
density chosen for εi.

The matrix X is the n by p matrix of explanatory variables. It is assumed that X is of rank p and that
a column or a linear combination of columns of X is not equal to the column vector of 1’s or a multiple
of it. This means that a constant term cannot be included in the model (1). The statistics a and M are
found as follows. Let εi have pdf f(ε) and let g = −f ′/f . Let W1,W2, . . . ,Wn be order statistics for a
random sample of size n with the density f(.). Define Zi = g(Wi), then ai = E(Zri

). To define M we
need M−1 = XT (B − A)X , where B is an n by n diagonal matrix with Bii = E(g

′(Wri
)) and A is a

symmetric matrix with Aij = cov(Zri
, Zrj

). In the case of the Normal distribution, the Z1 < . . . < Zn

are standard Normal order statistics and E(g′(Wi)) = 1 for i = 1, 2, . . . , n.

The analysis can also deal with ties in the data. Two observations are adjudged to be tied if
|Yi − Yj | < TOL, where TOL is a user-supplied tolerance level.
Various statistics can be found from the analysis:

(a) The score statistic XTa. This statistic is used to test the hypothesis H0 : β = 0, see (e).
(b) The estimated variance-covariance matrix XT (B −A)X of the score statistic in (a).
(c) The estimate β̂ =MXTa.
(d) The estimated variance-covariance matrix M = (XT (B −A)X)−1 of the estimate β̂.
(e) The χ2 statistic Q = β̂TM−1β̂ = aTX(XT (B − A)X)−1XTa used to test H0 : β = 0. Under H0,

Q has an approximate χ2 distribution with p degrees of freedom.

(f) The standard errors M
1
2
ii of the estimates given in (c).

(g) Approximate z-statistics i.e., Zi = β̂i/s.e.(β̂i) for testing H0 : βi = 0. For i = 1, 2, . . . , n, Zi has an
approximate N(0, 1) distribution.

In many situations, more than one sample of observations will be available. In this case we assume the
model,

hk(Yk) = X
T
k β + ek, k = 1, 2, . . . ,NS

where NS is the number of samples. In an obvious manner, Yk and Xk are the vector of observations
and the design matrix for the kth sample respectively. Note that the arbitrary transformation hk can be
assumed different for each sample since observations are ranked within the sample.

The earlier analysis can be extended to give a combined estimate of β as β̂ = Dd, where

D−1 =
NS∑
k=1

XT
k (Bk −Ak)Xk

and

d =
NS∑
k=1

XT
k ak

with ak, Bk and Ak defined as a, B and A above but for the kth sample.

The remaining statistics are calculated as for the one sample case.

4 References

[1] Pettitt A N (1982) Inference for the linear model using a likelihood based on ranks J. Roy. Statist.
Soc. Ser. B 44 234–243

5 Parameters

1: NS — INTEGER Input

On entry: the number of samples.

Constraint: NS ≥ 1.

2: NV(NS) — INTEGER array Input

On entry: the number of observations in the ith sample, for i = 1, 2, . . . ,NS.

Constraint: NV(i) ≥ 1, for i = 1, 2, . . . ,NS.

3: NSUM — INTEGER Input

On entry: the total number of observations.

Constraint: NSUM =
NS∑
i=1

NV(i).

4: Y(NSUM) — real array Input

On entry: the observations in each sample. Specifically, {Y
(

i−1∑
k=1

NV(k) + j

)
must contain the jth

observation in the ith sample.

5: IP — INTEGER Input

On entry: the number of parameters to be fitted.

Constraint: IP ≥ 1.

6: X(NX,IP) — real array Input

On entry: the design matrices for each sample. Specifically, X

(
i−1∑
k=1

NV(k) + j, l

)
must contain the

value of the lth explanatory variable for the jth observation in the ith sample.

Constraint: X must not contain a column with all elements equal.

G08RAF.2 [NP3390/19/pdf]



G08 – Nonparametric Statistics G08RAF

7: NX — INTEGER Input

On entry: the first dimension of the array X as declared in the (sub)program from which G08RAF
is called.

Constraint: NX ≥ NSUM.

8: IDIST — INTEGER Input

On entry: the error distribution to be used in the analysis as follows:

IDIST = 1 – Normal

IDIST = 2 – Logistic

IDIST = 3 – Extreme value

IDIST = 4 – Double-exponential

Constraint: IDIST = 1, 2, 3 or 4.

9: NMAX — INTEGER Input

On entry: the value of the largest sample size.

Constraint: NMAX = max
1≤i≤NS

(NV(i)) and NMAX > IP.

10: TOL — real Input

On entry: the tolerance for judging whether two observations are tied. Thus, observations Yi and
Yj are adjudged to be tied if |Yi − Yj | < TOL.

Constraint: TOL > 0.0.

11: PARVAR(NPVAR,IP) — real array Output

On exit: the variance-covariance matrices of the score statistics and the parameter estimates; the
former being stored in the upper triangle and the latter in the lower triangle. Thus for 1 ≤ i ≤ j ≤
IP, PARVAR(i, j) contains an estimate of the covariance between the ith and jth score statistics.
For 1 ≤ j ≤ i ≤ IP−1, PARVAR(i + 1, j) contains an estimate of the covariance between the ith
and jth parameter estimates.

12: NPVAR — INTEGER Input

On entry: the first dimension of the array PARVAR as declared in the (sub)program from which
G08RAF is called.

Constraint: NPVAR ≥ IP + 1.

13: IRANK(NMAX) — INTEGER array Output

On exit: for the one sample case, IRANK contains the ranks of the observations.

14: ZIN(NMAX) — real array Output

On exit: for the one sample case, ZIN contains the expected values of the function g(.) of the order
statistics.

15: ETA(NMAX) — real array Output

On exit: for the one sample case, ETA contains the expected values of the function g′(.) of the order
statistics.

16: VAPVEC(NMAX∗(NMAX+1)/2) — real array Output

On exit: for the one sample case, VAPVEC contains the upper triangle of the variance-covariance
matrix of the function g(.) of the order statistics stored column-wise.

17: PAREST(4∗IP+1) — real array Output

On exit: the statistics calculated by the routine as follows. The first IP components of PAREST
contain the score statistics. The next IP elements contain the parameter estimates. PAREST(2×
IP+1) contains the value of the χ2 statistic. The next IP elements of PAREST contain the standard
errors of the parameter estimates. Finally, the remaining IP elements of PAREST contain the
z-statistics.

18: WORK(LWORK) — real array Workspace
19: LWORK — INTEGER Input

On entry: the dimension of the array WORK as declared in the (sub)program from which G08RAF
is called.

Constraint: LWORK ≥ NMAX × (IP+1).

20: IWA(NMAX) — INTEGER Workspace

21: IFAIL — INTEGER Input/Output

On entry: IFAIL must be set to 0, −1 or 1. For users not familiar with this parameter (described
in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors detected by the routine:

IFAIL = 1

On entry, NS < 1,
or TOL ≤ 0.0,
or NMAX ≤ IP,
or NPVAR < IP + 1,
or NX < NSUM,
or NMAX �= max

1≤i≤NS
(NV(i)),

or NV(i) ≤ 0 for some i, NV(i),

or NSUM �=
NS∑
i=1

NV(i),

or IP < 1,
or LWORK < NMAX× (IP + 1).

IFAIL = 2

On entry, IDIST < 1,
or IDIST > 4.

IFAIL = 3

On entry, all the observations are adjudged to be tied. The user is advised to check the value
supplied for TOL.

IFAIL = 4

The matrix XT (B −A)X is either ill-conditioned or not positive-definite. This error should only
occur with extreme rankings of the data.

IFAIL = 5

The matrix X has at least one of its columns with all elements equal.

7 Accuracy

The computations are believed to be stable.

8 Further Comments

The time taken by the routine depends on the number of samples, the total number of observations and
the number of parameters fitted.

In extreme cases the parameter estimates for certain models can be infinite, although this is unlikely to
occur in practice. See Pettitt [1] for further details.

9 Example

A program to fit a regression model to a single sample of 20 observations using two explanatory variables.
The error distribution will be taken to be logistic.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* G08RAF Example Program Text
* Mark 14 Revised. NAG Copyright 1989.
* .. Parameters ..

INTEGER NIN, NOUT
PARAMETER (NIN=5,NOUT=6)
INTEGER NSMAX, NX, NMXMAX, NSMMAX, IPMAX, NPVAR, LWORK
PARAMETER (NSMAX=5,NX=100,NMXMAX=NX,NSMMAX=NX,IPMAX=6,

+ NPVAR=IPMAX+1,LWORK=NMXMAX*(IPMAX+1))
* .. Local Scalars ..

real TOL
INTEGER I, IDIST, IFAIL, IP, J, NMAX, NS, NSUM

* .. Local Arrays ..
real ETA(NMXMAX), PAREST(4*IPMAX+1),

+ PARVAR(NPVAR,IPMAX), VAPVEC(NMXMAX*(NMXMAX+1)/2),
+ WORK(LWORK), X(NX,IPMAX), Y(NSMMAX), ZIN(NMXMAX)
INTEGER IRANK(NMXMAX), IWA(NMXMAX), NV(NSMAX)

* .. External Subroutines ..
EXTERNAL G08RAF

* .. Intrinsic Functions ..
INTRINSIC MAX

* .. Executable Statements ..
WRITE (NOUT,*) ’G08RAF Example Program Results’

* Skip heading in data file
READ (NIN,*)

* Read number of samples, number of parameters to be fitted,
* error distribution parameter and tolerance criterion for ties.

READ (NIN,*) NS, IP, IDIST, TOL
WRITE (NOUT,*)
IF (NS.GT.0 .AND. NS.LE.NSMAX .AND. IP.GT.0 .AND. IP.LE.IPMAX)

+ THEN
WRITE (NOUT,99999) ’Number of samples =’, NS
WRITE (NOUT,99999) ’Number of parameters fitted =’, IP
WRITE (NOUT,99999) ’Distribution =’, IDIST
WRITE (NOUT,99998) ’Tolerance for ties =’, TOL

* Read the number of observations in each sample.
READ (NIN,*) (NV(I),I=1,NS)

NMAX = 0
NSUM = 0
DO 20 I = 1, NS

NSUM = NSUM + NV(I)
NMAX = MAX(NMAX,NV(I))

20 CONTINUE
IF (NMAX.GT.0 .AND. NMAX.LE.NMXMAX .AND. NSUM.GT.0 .AND.

+ NSUM.LE.NSMMAX) THEN
* Read in observations and design matrices for each sample.

READ (NIN,*) (Y(I),(X(I,J),J=1,IP),I=1,NSUM)
IFAIL = 0

*
CALL G08RAF(NS,NV,NSUM,Y,IP,X,NX,IDIST,NMAX,TOL,PARVAR,

+ NPVAR,IRANK,ZIN,ETA,VAPVEC,PAREST,WORK,LWORK,
+ IWA,IFAIL)

*
WRITE (NOUT,*)
WRITE (NOUT,*) ’Score statistic’
WRITE (NOUT,99997) (PAREST(I),I=1,IP)
WRITE (NOUT,*)
WRITE (NOUT,*) ’Covariance matrix of score statistic’
DO 40 J = 1, IP

WRITE (NOUT,99997) (PARVAR(I,J),I=1,J)
40 CONTINUE

WRITE (NOUT,*)
WRITE (NOUT,*) ’Parameter estimates’
WRITE (NOUT,99997) (PAREST(IP+I),I=1,IP)
WRITE (NOUT,*)
WRITE (NOUT,*) ’Covariance matrix of parameter estimates’
DO 60 I = 1, IP

WRITE (NOUT,99997) (PARVAR(I+1,J),J=1,I)
60 CONTINUE

WRITE (NOUT,*)
WRITE (NOUT,99996) ’Chi-squared statistic =’,

+ PAREST(2*IP+1), ’ with’, IP, ’ d.f.’
WRITE (NOUT,*)
WRITE (NOUT,*) ’Standard errors of estimates and’
WRITE (NOUT,*) ’approximate z-statistics’
WRITE (NOUT,99995) (PAREST(2*IP+1+I),PAREST(3*IP+1+I),I=1,

+ IP)
END IF

END IF
STOP

*
99999 FORMAT (1X,A,I2)
99998 FORMAT (1X,A,F8.5)
99997 FORMAT (1X,2F9.3)
99996 FORMAT (1X,A,F9.3,A,I2,A)
99995 FORMAT (1X,F9.3,F14.3)

END

9.2 Program Data

G08RAF Example Program Data
1 2 2 0.00001
20
1.0 1.0 23.0
1.0 1.0 32.0
3.0 1.0 37.0
4.0 1.0 41.0
2.0 1.0 41.0
4.0 1.0 48.0
1.0 1.0 48.0
5.0 1.0 55.0
4.0 1.0 55.0
4.0 0.0 56.0
4.0 1.0 57.0
4.0 1.0 57.0
4.0 1.0 57.0
1.0 0.0 58.0
4.0 1.0 59.0
5.0 0.0 59.0
5.0 0.0 60.0
4.0 1.0 61.0
4.0 1.0 62.0
3.0 1.0 62.0

9.3 Program Results

G08RAF Example Program Results

Number of samples = 1
Number of parameters fitted = 2
Distribution = 2
Tolerance for ties = 0.00001

Score statistic
-1.048 64.333

Covariance matrix of score statistic
0.673
-4.159 533.670

Parameter estimates
-0.852 0.114

Covariance matrix of parameter estimates
1.560
0.012 0.002

Chi-squared statistic = 8.221 with 2 d.f.

Standard errors of estimates and
approximate z-statistics

1.249 -0.682
0.044 2.567

